

Embedded System Design Lab 3

Stephen A. Edwards

Due August 8, 2005

Abstract

Reverse-engineer some synthesizable VHDL circuit models. Examine their source to draw a block diagram. Use a simulator to observe their behavior and draw timing diagrams. Explain what each circuit does.

1 Introduction

In this class, you'll be using VHDL (VHSIC Hardware Description Language) to describe hardware. It is a fairly verbose language, but fairly simple at its core. You will want to consult the *Writing VHDL for RTL Synthesis* handout available on the class webpage for examples of how to write VHDL. Unfortunately, the language is very big and large parts of it cannot be directly translated into hardware. As such, things like the language reference manual and the majority of VHDL books are useless because they are very complicated and it is difficult to determine when you may actually use what they teach to specify hardware.

Although the syntax of VHDL vaguely resembles that of an imperative language like C, do not be deceived: *VHDL is not a programming language*. In particular, the sort of imperative, algorithmic thinking that works well to solve problems in C *will not* work in VHDL. A C-like VHDL program will probably not compile, if it does compile, it probably will not work, and even if it does compile, you will not like the result.

VHDL is mostly a structural language. VHDL is useful mostly for defining how components connect. The main idea is that a system is composed of hierarchically-arranged blocks called entity/architecture pairs (everything in VHDL has a weird name). For each block, you define its interface (a list of wires that enter and leave it) and its guts, which may consist of instances of other blocks, dataflow expressions (e.g., a particular signal is the logical AND of two others), and processes that appear to contain imperative code.

2 An Example

Figure 1 shows a VHDL circuit model modeling a simple, and probably useless, ALU-like object. Defined in the entity, its inputs are a clock, reset and enable signals, and two eight-bit numbers. In response, it produces an eight-bit result. The architecture defines an eight-bit intermediate result called w and contains a single clock-triggered process that represents a block of combinational logic feeding a bank of eight edge-sensitive flip-flops (the w signal).

```
library ieee;
use ieee.std_logic_1164.all;
use ieee.std_logic_arith.all;
use ieee.std_logic_unsigned.all;

entity dumb_alu is
  port(clk      : in  std_logic;
        reset   : in  std_logic;
        enable  : in  std_logic;
        x, y    : in  std_logic_vector(7 downto 0);
        z      : out std_logic_vector(7 downto 0));
end dumb_alu;

architecture behavioral of dumb_alu is

  signal w : std_logic_vector(7 downto 0);

begin
  process (clk)
  begin
    if clk'event and clk='1' then
      if reset='1' then
        w <= "00000000";
      elsif enable='1' then
        w <= y - 4;
      else
        w <= x + y;
      end if;
    end if;
  end process;

  z <= w;
end behavioral;
```

Figure 1: A synthesizable VHDL entity/architecture.

The `clk'event and clk = '1'` idiom (a standard one) indicates that these are a positive-edge-triggered flip-flops. Inside this block is a series of if-then-else statements that assign to w depending on whether *reset* is true and whether *enable* is true. Together, these imply the steering (multiplexer) and arithmetic logic shown in Figure 3.

The semantics of VHDL are such that the inputs to this block are read just before the rising edge of the clock and the outputs are produced just after the rising edge, as shown in Figure 4. Note that the inputs may change at any time during a clock signal but that the output only changes on the rising edge.

```

library ieee;
use ieee.std_logic_1164.all;
use ieee.std_logic_arith.all;
use ieee.std_logic_unsigned.all;

entity testbench is
end testbench;

architecture behavioral of testbench is

    signal clk : std_logic := '0';
    signal reset, enable : std_logic;
    signal x, y, z: std_logic_vector(7 downto 0);

    component dumb_alu
    port(
        clk      : in  std_logic;
        reset    : in  std_logic;
        enable   : in  std_logic;
        x, y     : in  std_logic_vector(7 downto 0);
        z       : out std_logic_vector(7 downto 0));
    end component;

begin

    device_under_test: dumb_alu
    port map (
        clk => clk,
        reset => reset,
        enable => enable,
        x => x,
        y => y,
        z => z);

    clkgen : process
    begin
        wait for 10 ns;
        clk <= not clk;
    end process;

    reset <= '1',
            '0' after 60 ns;

    enable <= '0',
             '1' after 120 ns,
             '0' after 140 ns,
             '1' after 160 ns,
             '0' after 180 ns;

    datagen : process
    begin
        wait for 51 ns;
        x <= X"04"; y <= X"02"; wait for 20 ns;
        x <= X"03"; y <= X"00"; wait for 20 ns;
        x <= X"08"; y <= X"01"; wait for 20 ns;
        x <= X"07"; y <= X"04"; wait for 20 ns;
        x <= X"05"; y <= X"03"; wait for 20 ns;
        x <= X"06"; y <= X"05"; wait for 40 ns;
        x <= X"07"; y <= X"01";
        wait; -- forever
    end process;

end behavioral;

```

Figure 2: A testbench for Figure 1.




Figure 3: A block diagram for Figure 1.

Figure 2 shows a testbench for the block in Figure 1. Its job is to provide stimulus to the block so that a simulation does something interesting. This particular testbench applies the waveforms shown in Figure 4 to produce Figure 6, a screenshot from the Cadence waveform viewer that we will use to observe the output of these modules.

The testbench uses a different subset of the VHDL language that is not synthesizable, i.e., cannot automatically be turned into logic. For example, the clock is generated by the *clkgen* process in Figure 2 that toggles the signal every 10 ns, a behavior that cannot be implemented using only logic gates (you need a controlled oscillator).

A clock generator is probably the most common part of a testbench. Other parts include assignment statements using the *after* keyword, which provides control over the time at which the assignment will take place (c.f., the statements generating the *reset* and *enable* signals), and processes with delays for generating more complex waveforms. The *datagen* process is an example of this, which first waits until slightly after the second clock cycle before generating a sequence of assignments to the *x* and *y* vectors. Notice the use of a final, lone *wait* statement, which effectively terminates the process (it will otherwise repeat indefinitely).

I have created four projects in Xilinx's Project Navigator, part of the ISE environment designed for hardware synthesis (the EDK that we have been using uses ISE as a back-end). Opening these projects (the files named *.npl in the lab3.zip file) will open Project Navigator and let you browse the source code and run the ModelSim VHDL simulator.

To run the simulator, click on *test_bench.vhd* and choose "Simulate Behavioral Model" under "ModelSim Simulator" in the "Process" window below the list of source files. This will run the simulation and open three windows.

The most interesting window is the "wave" window, which shows an oscilloscope-like display of the values of the signals in the design. The default zoom seems to be useless: use the zoom button and scroll bars to navigate.

3 The Assignment

Draw block and timing diagrams for the three VHDL designs (design1, design2, and design2) in the *lab3.zip* file and explain what they do. The lab-example directory contains the example and test bench in this lab handout.

Use the ModelSim simulator to validate your analysis of the circuits. Write a testbench for each that exercises the design and illustrates its operation. I have already written simple test benches for you that need to be extended.

Show the TA the waveforms on the screen and hand in your block and timing diagrams for each of these three designs.




Figure 5: The ALU project in Project Navigator.




Figure 6: The ModelSim windows from simulating the ALU project.